# Maintainer: Lukas Fleischer <lfleischer@archlinux.org>
# Contributor: Paul Mattal <paul@archlinux.org>
# Contributor: Tom Newsom <Jeepster@gmx.co.uk>

pkgname=enscript
pkgver=1.6.6
pkgrel=2
pkgdesc="Convert ASCII files to PostScript suitable for printing"
arch=('x86_64')
url="http://git.savannah.gnu.org/cgit/enscript.git"
license=('GPL3')
depends=('glibc')
backup=('etc/enscript/enscript.cfg')
source=(ftp://ftp.gnu.org/gnu/$pkgname/$pkgname-$pkgver.tar.gz)
md5sums=('3acc242b829adacabcaf28533f049afd')

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --sysconfdir=/etc/enscript
  make
}

check() {
  cd $pkgname-$pkgver
  make check
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
